NASA CONTRACTOR
REPORT

NASA CR-120610

THE EFFECTS OF CORRELATION ON GOODNESS OF FIT

By Larry J. Kitchens
Department of Mathematical Sciences
Appalachian State University
Boone, North Carolina 28607

January 6, 1975

Prepared for

NASA-GEORGE C. MARSHALL SPACE FLIGHT CENTER
Marshall Space Flight Center, Alabama 35812

1. REPORT NO.	2. GOVERNMENT ACCESSION NO.	3. RECIPIENT'S CATALOG NO.
NASA CR-120610		

4. TITLE AND SUBTITLE	5. REPORT DATE
The Effects of Correlation on Goodness of Fit	January 6, 1975
	6. PERFORMING ORGANIZATION CODE
7. AUTHOR(S) Larry J. Kitchens	8. PERFORMING ORGANIZATION REPORT #
9. PERFORMING ORGANIZATION NAME AND ADDRESS Department of Mathematical Sciences Appalachian State University Boone, North Carolina 28607	10. WORK UNIT NO. 11. CONTRACT OR GRANT NO. NAS8-29286
12. SPONSORING AGENCY NAME AND ADDRESS National Aeronautics and Space Administration Washington, DC 20546	13. TYPE OF REPORT & PERIOD COVERED Contractor 14. SPONSORING AGENCY CODE

15. SUPPLEMENTARY NOTES

Prepared under the technical direction of the Aerospace Environment Division, Space Sciences Laboratory, Marshall Space Flight Center; Contract monitor is Mr. Lee W. Falls.

16. ABSTRACT

In this paper we generate autocorrelated normal random variates via computer and study the effects of various levels of correlation on goodness of fit problems. The results will be useful in determining the distribution or estimating the parameters of populations in which we have correlated observations such as wind speeds and temperature. The model used to generate the autocorrelated data is an autoregressive process of order 1. The Kolmogorov-Smirnov and chi-square statistics are used in the analysis.

It was observed in the simulation that high positive correlations tend to shift the sample mean away from the population mean and negative correlations tend to shift the sample mean towards the population mean. In many cases, it was observed that positive and negative correlations tend to decrease the standard deviation. However, since this did not occur in all cases, no definite conclusion could be made regarding the standard deviation.

Since the autoregressive process is a linear transformation, it was not suprising that normality was preserved. However, a possible extension of this problem would be to generate non-normal data and observe how the distribution is affected by correlation. Another extension would be to utilize another model such as autogregressive of order k or a moving average process of order k.

17. KEY WORDS	18. DISTRIBUTION STATEMENT
Statistical Analysis Autocorrelation Goodness of Fit Tests Kolmogorov-Smirnov Test Chi-Square Test Autoregressive Processes	Unclassified-Unlimited James D Ledbetter for Lee W. Falls

19. SECURITY CLASSIF. (of this report)	20. SECURITY CLASSIF. (of this page)	21. NO. OF PAGES	22. PRICE
Unclassified	Unclassified	37	NTIS

MSFC - Form 3292 (Rev December 1972)

For sale by National Technical Information Service, Springfield, Virginia 22151

I. INTRODUCTION

The problem considered in this paper is to generate auto-correlated random variates via computer and study the effects of various levels of correlation on goodness of fit problems. This is of interest to the Terrestrial Environment Branch, Aerospace Environment Division, Space Sciences Laboratory, George C. Marshall Space Flight Center, Alabama, and the financial support for the project was under NASA contract number NAS8-29286. The results will be useful in determining the distribution or estimating the parameters of populations in which we have correlated observations, such as wind speed and temperature.

The model and simulation techniques are presented in section 2. Utilizing a Univac 70/46 computer, normal correlated data is generated and analyzed in section 3. The Kolmogorov-Smirnov statistic and chi-square goodness of fit are used in the analysis. Section 4 describes the computer programs and their use with the actual programs listed in the appendix.

2. MODEL AND SIMULATION

If X and Y are jointly normally distributed, then it is known that $Y|X$ is normally distributed with mean

$\mu_y + \rho\frac{\sigma_y}{\sigma_x}(X - \mu_x)$ and variance $\sigma^2_y(1 - \rho^2)$. Assume $\mu_x = \mu_y = \mu$ and $\sigma_x = \sigma_y = \sigma$ then

$$Y|X \sim N(\mu + \rho\ (X - \mu),\ \sigma^2(1 - \rho^2))$$

or

$$Y|X = \mu + \rho(X - \mu) + \sigma\sqrt{1 - \rho^2}\cdot Z$$

where $Z \sim N(0,1)$.

Thus it was decided to recursively generate a random sample, $X_1,\ldots,X_m$ via the formula

$$X_i = \mu + \rho\ (X_{i-1} - \mu) + \sigma\sqrt{1 - \rho^2}\ Z_i \tag{2.1}$$

where $Z_i \sim N(0,1)$.

By mathematical induction it can be shown that

$$E[X_i] = \mu \quad \text{and} \quad Var[X_i] = \sigma^2 \quad \text{for all } i. \tag{2.2}$$

Moreover, with a little patience, the auto-correlation function is given by

$$\rho_{xx}(\tau) = \rho^{|\tau|} \tag{2.3}$$

Thus with (2.1) we are able to generate a sequence of autocorrelated normal variates.

After searching the literature it was learned that (2.1) is a version of a first order autoregressive process. Mihram [3] defines an autoregressive process $X(t)$ of order m by

$$X(t) - \mu_x = \sum_{i=1}^{m} a_i [X(t-i) - \mu_x] + \varepsilon(t)$$

where $\{\varepsilon(t)\}$ is a white-noise process of mean zero. Thus we could use (2.1) to generate variates of any specified distribution with autocorrelation given in (2.3), by choosing the white-noise appropriately.

One other process that might be useful in applications, particularly to wind speeds mentioned by Falls [1], is the moving average process of order k. This model is of the form

$$X(t) - \mu_x = \sum_{i=0}^{k} h_i \varepsilon(t-i) \tag{2.4}$$

where $\{\varepsilon(t)\}$ is a white noise process of mean zero. The advantage of this model is that the autocorrelation is zero when the lag τ exceeds k. For example, the second-order moving average process

$$X(t) - \mu = \varepsilon(t) + .75\varepsilon(t-1) + .25\varepsilon(t-2)$$

has autocorrelation function

$$\rho_{xx}(\tau) = \begin{cases} 1 & \tau = 0 \\ .577 & \tau = 1 \\ .154 & \tau = 2 \\ 0 & \tau \geq 3 \end{cases}$$

Thus the correlation goes to zero after each third observation. The model given in (2.1) was simulated by first generating k uniformly distributed random numbers in (0,1). By the Central

Limit Theorem it is known that as $k \to \infty$, $Z = (\sum_{i=1}^{k} X_i - k/2)/\sqrt{k/12}$ approaches a standard normal distribution. It is standard procedure to choose $k = 12$; thus, for each 12 uniformly distributed random numbers we obtained a single realization from a standard normal population. By this technique samples of size $n = 20, 40, 100$, and 500 were generated. Each sample, $\{Z_i : i=1,\ldots,n\}$, was transformed via (2.1) to yield our autocorrelated samples $\{X_i : i=1,\ldots,n\}$. Correlations of $\pm .1$, $\pm .3$, $\pm .5$, $\pm .7$ and $\pm .9$ were used.

3. RESULTS-GOODNESS OF FIT

To simplify the problem, a mean of zero and standard deviation of one were chosen in all analyses. For each sample size, five distinct samples were generated from the $N(0,1)$ population. These samples were then checked for fit by the Kolmogorov-Smirnov and chi-square statistics before and after applying the transformation. An example is presented in Table 3.1. Naturally, for small sample sizes the fit of the initial data was not very good in many cases. This was due to the fact that the sample mean was not close to zero or the sample standard deviation was not close to one. Be that as it may, it was observed that after applying the transformation (2.1) small correlations did not affect the fit as much as large positive correlations. This is pointed out in figures 3.2 through 3.6. In each instance the probability level of the K-S statistic is plotted before and after the transformation. The probability before transforming is represented by a dot and by a cross after transforming. It is observed that for $\rho = .1$ there is not much difference in the probabilities; however, for $\rho = .9$ there is a large difference and in many cases the probabilities approach zero after transforming. Figures 3.7 through 3.10 illustrate the probabilities for negative correlations. It is interesting to note from figure 3.10 that a bad fit was corrected by a large negative correlation. The reason for this will be given later.

Mean = 0.0   Standard Deviation = 1.0   Correlation = 0.9   Number = 100

	Sample 1		Sample 2		Sample 3		Sample 4		Sample 5	
	Normal	Corre-lated	Normal	Corre-lated	Normal	Corre-lated	Normal	Corre-lated	Normal	Corre-lated
Average	0.1977	0.8578	-0.0885	-0.4165	-0.0303	-0.0356	0.0039	0.0245	0.0330	0.1778
Std. Dev.	1.0137	0.8395	1.0037	1.0033	0.9624	0.8244	0.9030	0.7392	0.9939	1.1959
K-S Stat.	0.1081	0.3849	0.0853	0.1598	0.0717	0.1086	0.0766	0.0990	0.0578	0.1451
Prob. of	0.1930	0.0000	0.4609	0.0121	0.6821	0.1886	0.6009	0.2808	0.8917	0.0296
Chi-Sq.	12.6000	88.6000	7.4000	20.4000	4.000	14.8000	13.2000	14.2000	4.6000	27.6000
DF	99	99	99	99	99	99	99	99	99	99

TABLE 3.1



Figure 3.2 Probabilities of K-S Statistic When Corr.=0.1



Figure 3.3 Probabilities of K-S Statistic When Corr.= 0.3



Figure 3.4 Probabilities of K-S Statistic when Corr.= 0



Figure 3.5 Probabilities of K-S Statistic When Corre.= 0.7



Figure 3.6 Probabilities of K-S Statistic When Corre.= 0.9



Figure 3.7 Probabilities of K-S Statistic When Corr. = .041



Figure 3.8 Probabilities of K-S Statistic When Corr.= -0.5



Figure 3.9 Probabilities of K-S Statistic When Corr.= -0.7



Figure 3.10 Probabilities of K-S Statistic When Corr.= -0.9

The next step in the problem was to generate a sample from the N(0,1) population, apply the transformation five time using the correlations of .1, .3, .5, .7 and .9 to observe, in addition to checking the goodness of fit, how the means and standard deviations were affected. An example of the output is presented in Table 3.11.

	Mean=0.0		Std.Dev.=1.0		N=100	
	Normal	Corr.=0.1	Corr.=0.3	Corr.=0.5	Corr.=0.7	Corr.=0.9
Average	0.0039	0.0046	0.0059	0.0073	0.0103	0.0246
St. Dev.	0.9030	0.9023	0.8934	0.8695	0.8263	0.7392
K-S Stat.	0.0766	0.0933	0.0794	0.0688	0.0900	0.0990
Prob. of	0.6009	0.3489	0.5535	0.7319	0.3923	0.2808
Chi-Sq.	13.2	6.2	17.4	8.0	8.6	14.2
DF	99	99	99	99	99	99

TABLE 3.11

Theoretically, from (2.2) we would expect the mean and standard deviation to remain constant for all levels of correlation. However, this is not the case since we are sampling from a N(0,1) population and the sample mean will not necessarily be zero nor will the sample standard deviation be one. From Table 3.11 we see that initially the sample mean is positive and after applying increasing levels of correlation the sample mean becomes even more positive. Had the initial sample mean been negative it would have become more negative as the level of correlation increased. There were five samples of size 20, five of size 40, five of size 100 and five of size 500 generated, and in every case where the sample size was 40 or greater the same shift in sample means was observed. That is, applying a positive correlation tended to shift the sample mean away from zero.

We also observe from Table 3.11 that the standard deviation was decreased. Although this did not happen every time, in those cases where the standard deviation increased it only increased slightly. One conjecture that might be made is that observations on the tails of the distribution are affected more drastically than those close to the mean. More specifically, they are shifted towards the mean by the higher correlations resulting in a smaller spread.

Initially, it was believed that negative correlations would yield the same results. However, after investigating (five samples of size 20, ten of size 40, ten of size 100 and five of size 500) it was learned that negative correlations tend to shift the sample mean the opposite direction towards zero and in a few cases even past zero. An example of this is presented in Table 3.12.

Mean=0.0 Std.Dev.=1.0 N=100

	Normal	Corr.=-.1	Corr.=-.3	Corr.=-.5	Corr.=-.7	Corr.=-.9
Average	0.0680	0.0616	0.0496	0.0381	0.0262	0.0097
St. Dev.	1.1046	1.0993	1.0902	1.0739	1.0243	0.9044
K-S Stat.	0.0897	0.0751	0.0729	0.0615	0.0478	0.0963
Prob. of	0.3963	0.6260	0.6629	0.8434	0.9763	0.3122
Chi-Sq.	8.8000	6.2000	19.2000	7.2000	3.8000	7.2000
DF	99	99	99	99	99	99

TABLE 3.12

As in the case for positive correlations, negative correlations tend to decrease the standard deviation. However, since this did not occur in all cases, no definite conclusion could be made.

In every case studied a histogram of the data was printed and it appeared that in all cases normality was preserved. This should be expected since normality is preserved under a linear transformation. Thus, the bad fits recorded by the K-S statistic was apparently due to the fact that the sample mean and standard deviation were not close to zero and one respectively. Since negative correlations shift the mean towards zero, this would explain why high negative correlation would tend to correct the data and give a better fit to N(0,1).

A possible extension of the problem would be to test the goodness of fit by the K-S statistic using $\overline{X}$ and S for the mean and standard deviation and utilize the tables constructed by Lilliefors [2].

Other extensions of the problem would be to generate non-normal data and observe how the goodness of fit is affected by correlations or utilize the moving average process of order k given in (2.4) with normal and non-normal data.

4. COMPUTER PROGRAM USAGE

The main line program is titled NASAI and utilizes seven different subroutines. Subroutine RANDU is a standard random number generator. Subroutine KOLMO which uses subroutines NDTR and SMIRN computes the Kolmogorov-Smirnov statistic and the probability of exceeding the observed value of that statistic. Subroutine CHISQR computes the chi-square statistic. Subroutine FREQUE calculates frequencies for input for HISTO which computes a histogram of the data.

To use the programs you must supply two data cards:

CARD 1

Col 1-10: A large odd integer right justified for a seed of the random number generator

Col 11-14: The size of the sample you wish to generate ($1 \leq N \leq 9999$, right justified)

Col 15: The number of distinct samples of size N you wish to generate ($1 \leq M \leq 9$)

CARD 2

The data may be transformed via equation 2.1 using five distinct correlations. The correlations (positive or negative) are placed in columns 5 through 24. Be sure to punch the decimal point or use a F4.1 format.

Col 1 - 4 : blank

Col 5 - 8 : first correlation

Col 9 - 12: second correlation

Col 13 - 16:	third correlation
Col 17 - 20:	fourth correlation
Col 21 - 24:	fifth correlation
Col 25 - 27:	The mean of the population you wish to generate your sample from. Punch the decimal point or use F3.0.
Col 28 - 30:	The standard deviation of the population you wish to generate your sample from. Punch the decimal point or use F3.0.

The output of the program will be a table similar to Table 3.11 followed by six histograms. The first is a histogram of the uncorrelated normal data. Then the next five histograms will be of the correlated data with the five different degrees of correlation. This procedure will be repeated the number of times you specified in column 15 of the first data card.

Summary

Several observations can be made from analyzing the data generated in this project. First, it should be observed that there is no observable distinction between samples of size 100 and samples of size 500. Thus we may assume that the process tends to converge after 100 iterations. Next, the degree of correlation built in did adversely affect the distribution. The higher the correlation between observations the more the data was transformed. This can be observed by investigating figures 3.2 through 3.6. The probability of the K-S statistic dropped significantly for larger positive correlations. This is due to the fact that positive correlations tend to shift the sample mean away from the population mean--the larger the correlation the greater the shift; and, in most instances the standard deviation became smaller. This is pointed out by Table 3.11. Although no statistical technique was used, it was observed from the histograms that normality was preserved, so the apparent lack of fit was due to the shift in means and change in the standard deviation.

For negative correlation the changes were not as drastic. The shift of the sample mean was towards the population mean rather than away as was the case for positive correlations. This is probably due to an oscillating effect. This also explains why in many cases a bad fit due to a shifted mean was corrected to give a better fit. The same change in standard deviation was observed. Also, a study of the histograms for the negative correlations indicates that the lack of fit was not due to a change in the distribution but rather a change in the parameters of the distribution.

References

1. Falls, Lee W., Personal Communication (November, 1974), rshall Space Flight Center, NASA, Alabama.

2. Lilliefors, Hubert W., "On the Kolmogorov-Smirnov Test for Normality With Mean and Variance Unknown," Journal of the American Statistical Association, 62, 399-402.

3. Mihram, G. Arthur, Simulation Statistical Foundations and Methodology, Academic Press, New York.

ACKNOWLEDGEMENTS

The author wishes to express his appreciation to O. E. Smith and L. W. Falls for their valuable comments and criticisms. Also, many thanks to Larry Bridges and Leland Jones for writing most of the computer programs used in this project.

APPENDIX

Program Listings

PRECEDING PAGE BLANK NOT FILMED

```
      PROGRAM NASAI
      REAL*8 YFL
      REAL X(6,1000),RO(6),SUMX(6),SUMXSQ(6),STDX(6),STA2(6)
     1,PR2(6),CHI2(6),XDF2(6),SUSUX(6),SUSTDX(6),FRE(14)
      DATA Y,KSTAR,IDASH,NSPCE/'Y','**','-',' '/
      NRO=6
  100 READ(5,1)IX,N,M
    1 FORMAT(I10,I4,I1)
      IF(IX.LE.0)STOP
      READ(5,12)RO,SI,CHI
   12 FORMAT(6F4.1,2F3.0)
      DO102L=1,6
      SUSUX(L)=0
      SUSTDX(L)=0
  102 CONTINUE
      DO11L=1,M
      WRITE(6,18)SI,CHI,N,RO
   18 FORMAT('0','MEAN=',F3.1,5X,'STD DEV=',F3.1,5X,
     1'N=',I4/,11X,6( 'COR=',F4.1,2X))
      DO16NN=1,N
   16 X(1,NN)=0.0
      DO99K=1,NRO
      SUMXSQ(K)=0.0
      SUMX(K)=0.
   99 STDX(K)=0
      DO3J=1,N
      DO2I=1,12
      CALL RANDU(IX,IY,YFL)
    2 X(1,J)=X(1,J)+YFL
      X(1,J)=X(1,J)-6
      DO3K=1,NRO
      IF(J.NE.1)GOTO5
      X(K,1)=X(1,1)
      GOTO7
    5 X(K,J)= SI+RO(K)*(X(K,J-1)-SI)+CHI*SQRT(1-(RO(K)**2))*(X(1,J))
    7 SUMX(K)=SUMX(K)+X(K,J)
      SUMXSQ(K)=SUMXSQ(K)+X(K,J)**2
    3 CONTINUE
      DO103K=1,NRO
      SUMX(K)=SUMX(K)/N
      STDX(K)=(SUMXSQ(K)-(N*(SUMX(K)**2)))
      STDX(K)=SQRT(STDX(K)/(N-1))
      SUSUX(K)=SUSUX(K)+SUMX(K)
      SUSTDX(K)=SUSTDX(K)+STDX(K)
      LLL=1
      CALL KOLMO(X,N,K,STA2,PR2,LLL,SI,CHI,IER)
  103 CALL CHISQR(X,N,K,CHI2,XDF2)
      WRITE(6,8)SUMX
    8 FORMAT(' ','AVERAGE',2X,6(F9.5,1X))
      WRITE(6,10)STDX
```

```
 10     FORMAT(' ','STD        ',6(F9.5,1X))
        WRITE(6,25)STA2
 25     FORMAT(' ','K-S STAT ',6(F9.5,1X))
        WRITE(6,47)PR2
 47     FORMAT(' ','PROB OF   ',6(F9.5,1X))
        WRITE(6,87)CHI2
 87     FORMAT(' ','CHI-SQ    ',6(F9.5,1X))
        WRITE(6,88)XDF2
 88     FORMAT(' ','DEGRE FRE',6(F9.5,1X)//)
        DO11KKK=1,6
        RRR=RO(KKK)
        CALL FREQUE(X,KKK,N,FRE)
        CALL HISTO(RRR,FRE,14,KSTAR,NSPCE,IDASH)
 11     CONTINUE
        DO104K=1,NRO
        SUSUX(K)=SUSUX(K)/N
 104    SUSTDX(K)=SUSTDX(K)/N
        WRITE(6,17)SUSUX,SUSTDX
 17     FORMAT('0','SUM AVRE.'6(F9.5,1X),/' STD AVRE.'
       1,6(F9.5,1X))
        GOTO100
        END
```

```
      SUBROUTINE RANDU(IX,IY,YFL)
      REAL*8 YFL
      IY=IX*65539
      IF(IY)5,6,6
    5 IY=IY+2147483647+1
    6 YFL=IY
      YFL=YFL*.4656613E-9
      IX=IY
      RETURN
      END
```

```
      SUBROUTINE KOLMO(X,N,K,Z,PROB,IFCOD,U,S,IER)
      DIMENSION X(6,1000),PROB(6),Z(6)
      IER=0
      DO 5 I=2,N
      IF(X(K,I)-X(K,I-1))1,5,5
1     TEMP=X(K,I)
      IM=I-1
      DO 3 J=1,IM
      L=I-J
      IF(TEMP-X(K,L))2,4,4
2     X(K,L+1)=X(K,L)
3     CONTINUE
      X(K,1)=TEMP
      GOTO5
4     X(K,L+1)=TEMP
5     CONTINUE
      NM1=N-1
      XN=N
      DN=0.0
      FS=0.0
      IL=1
6     DO7I=IL,NM1
      J=I
      IF(X(K,J)-X(K,J+1))9,7,9
7     CONTINUE
8     J=N
9     IL=J+1
      FI=FS
      FS=FLOAT(J)/XN
      IF(IFCOD-2)10,13,17
10    IF(S)11,11,12
11    IER=1
      GO TO 29
12    Z(K)=(X(K,J)-U)/S
      CALL NDTR(Z,Y,D,K)
      GOTO27
13    IF(S)11,11,14
14    Z(K)=(X(K,J)-U)/S+1.0
      IF(Z(K))15,15,16
15    Y=0.0
      GOTO27
16    Y=1.-EXP(-Z(K))
      GOTO27
17    IF(IFCOD-4)18,20,26
18    IF(S)19,11,19
19    Y=ATAN((X(K,J)-U)/S)*0.3183099+0.5
      GOTO27
20    IF(S-U)11,11,21
21    IF(X(K,J)-U)22,22,23
22    Y=0.0
```

```
      GOTO27
   23 IF(X(K,J)-S)25,25,24
   24 Y=1.0
      GO TO27
   25 Y=(X(K,J)-U)/(S-U)
      GOTO27
   26 IER=1
      GOTO29
   27 EI=ABS(Y-FI)
      ES=ABS(Y-FS)
      DN=AMAX1(DN,EI,ES)
      IF(IL-N)6,8,28
   28 Z(K)=DN*SQRT(XN)
      CALL SMIRN(Z,PROB,K)
      Z(K)=Z(K)/SQRT(XN)
      PROB(K)=1.0-PROB(K)
   29 RETURN
      END
```

```
      SUBROUTINE SMIRN(X,Y,K)
      DIMENSION X(6),Y(6)
      IF(X(K)-.27)1,1,2
    1 Y(K)=0.0
      GOTO9
    2 IF(X(K)-1.0)3,6,6
    3 G1=EXP(-1.233701/X(K)**2)
      G2=G1*G1
      G4=G2*G2
      G8=G4*G4
      IF(G8-1.0E-25)4,5,5
    4 G8=0.0
    5 Y(K)=(2.506628/X(K))*G1*(1.0+G8*(1.0+G8*G8))
      GOTO9
    6 IF(X(K)-3.1)8,7,7
    7 Y(K)=1.0
      GOTO9
    8 G1=EXP(-2.0*X(K)*X(K))
      G2=G1*G1
      G4=G2*G2
      G8=G4*G4
      Y(K)=1.0-2.0*(G1-G4+G8*(G1-G8))
    9 RETURN
      END
```

```
      SUBROUTINE NDTR(X,P,D,K)
      DIMENSION X(6)
      XX=X(K)
      AX=ABS(XX)
      T=1.0/(1.0+.2316419*AX)
      D=0.3989423*EXP(-XX*XX/2.0)
      P=1.0-D*T*((((1.330274*T-1.821256)*T+1.781478)*T-
     10.3565638)*T+0.3193815)
      IF(XX)1,2,2
    1 P=1.0-P
    2 RETURN
      END
```

```
      SUBROUTINE CHISQR(A,N,K,CHIVAL,XDF)
      DIMENSION A(6,1000),B(9),CELL(10),XDF(6),CHIVAL(6)
      DO 1 IX=1,10
    1 CELL(IX)=0.0
      CHIVAL(K)=0.0
      B(1)=-1.2817
      B(2)=-0.8418
      B(3)=-0.5244
      B(4)=-0.2533
      B(5)=0.0
      DO2J=6,9
    2 B(J)=-B(10-J)
      EXPVAL=FLOAT(N)/10.
      DO3 L=1,N
      IF(A(K,L).LT.B(1))CELL(1)=CELL(1)+1.
      IF(A(K,L).GT.B(9))CELL(10)=CELL(10)+1.
      DO 3 M=2,9
      IF(A(K,L).GT.B(M-1).AND.A(K,L).LE.B(M))CELL(M)=CELL(M)+1.
    3 CONTINUE
      DO5 I=1,10
    5 CHIVAL(K)=CHIVAL(K)+(((CELL(I)-EXPVAL)**2)/EXPVAL)
      XDF(K)=N-1
      RETURN
      END
```

```
      SUBROUTINE FREQUE(Y,NN,N,FR)
      DIMENSION Y(6,1000),FR(14),C(13)
      DO1I=1,14
    1 FR(I)=0.
      C(1)=-3.0
      DO2J=2,13
    2 C(J)=C(J-1)+0.5
      DO7K=1,N
      IF(Y(NN,K).LT.C(1))GOTO4
      IF(Y(NN,K).GE.C(13))GOTO5
      DO3L=2,13
      LK=L
      IF(Y(NN,K).GE.C(L-1).AND.Y(NN,K).LT.C(L))GOTO6
    3 CONTINUE
      GOTO7
    4 FR(1)=FR(1)+1.
      GOTO7
    5 FR(14)=FR(14)+1.
      GOTO7
    6 FR(LK)=FR(LK)+1.
    7 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE HISTO(RNU,FREQ,IN,K,NOTH,MINUS)
      DIMENSION JOUT(14),FREQ(14),IDSH(132)
      DIMENSION B(13)
      DO1I=1,132
    1 IDSH(I)=MINUS
      WRITE(6,4)RNU
    4 FORMAT('OCORR=',F5.1)
      DO12I=1,IN
   12 JOUT(I)=FREQ(I)
      WRITE(6,5)(JOUT(II),II=1,IN)
    5 FORMAT(' FREQUENCY',14I8)
      WRITE(6,7)IDSH
    7 FORMAT(1X,132A1)
      FMAX=0.0
      DO20I=1,IN
      IF(FREQ(I).LE.FMAX)GOTO20
      FMAX=FREQ(I)
   20 CONTINUE
      JSCAL=1
      IF(FMAX.LE.50.0)GOTO40
      JSCAL=(FMAX+49.0)/50.0
      WRITE(6,30)K,JSCAL
   30 FORMAT(' EACH',2X,A2,2X,'EQUALS ',I3,' POINTS'/)
   40 DO50I=1,IN
   50 JOUT(I)=NOTH
      MAX=FMAX/FLOAT(JSCAL)
      DO80I=1,MAX
      X=MAX-(I-1)
      DO70J=1,IN
      IF(FREQ(J)/FLOAT(JSCAL).LT.X)GOTO70
      JOUT(J)=K
   70 CONTINUE
      IX=X*FLOAT(JSCAL)
   80 WRITE(6,81)IX,(JOUT(J),J=1,IN)
   81 FORMAT(I6,5X,14(6X,A2))
      DO90I=1,IN
   90 JOUT(I)=I
      WRITE(6,7)IDSH
      B(1)=-3.0
      DO905IK=2,13
  905 B(IK)=B(IK-1)+0.5
      WRITE(6,91)(B(KKK),KKK=1,13)
   91 FORMAT(' INTERVAL',10X,13(F4.1,4X))
      RETURN
      END
```